

Demo22 ให้ copy demo 21 และเพิ่ม ปุ่ม 3 ปุ่ม คือ ปุ่ม Add ปุ่ม Update ปุ่มDelete

Object	Propety	Setting
Command Button	Name	CmdAdd
	Caption	&Add
	Name	CmdUpdate
	Caption	&Update
	Name	CmdDel
	Caption	&Delete

Method ที่ใช้ในการจัดการกับข้อมูลของ Data Control

1. datacontrol.Recordset.Addnew คือ Method ที่ใช้เพิ่มข้อมูลของ Data Control เพื่อสร้าง เนื้อที่ว่างสำหรับ record ใหม่ต่อท้ายใน recordSet

datacontrol.Recordset.Edit ใช้แก้ไขข้อมูลของ data control ก่อนที่จะบันทึกข้อมูล

2. datacontrol.Recordset.Update ใช้บันทึกข้อมูลของ data control เมื่อใช้ method AddNew และ Edit แล้ว ต้องใช้ method Update หลังจากที่ใช้ method Update แล้ว pointer จะถูกเลื่อนไปอยู่ที่ record แรกเสมอ ถ้าต้องการให้ record ที่แก้ไขยังคงปรากฏอยู่บนจอภาพต้องอาศัยค่า property bookmark และ lastModified เช่น

```
data1.Recordset.Update
Data1.Recordset.Bookmark = Data1.Recordset.LastModified
```

Bookmark ใช้บอกตำแหน่งของ pointer

LastModified ใช้บอกตำแหน่งของ record สุดท้ายที่เกิดการเปลี่ยนแปลงค่า

3. Datacontrol.Recordset.Delete ใช้ลบ record หลังจากลบ record แล้วควรขยับ pointer เนื่องจาก record ปัจจุบันได้ถูกลบไปแล้ว โดยใช้ method move

MoveFirst เลื่อน pointer ของข้อมูลไปยัง record แรกใน Recordset

MoveLast เลื่อน pointer ของข้อมูลไปยัง record สุดท้ายใน Recordset

MoveNext เลื่อน pointer ของข้อมูลไปยัง record ถัดไปของ record ปัจจุบันใน Recordset

MovePrevious เลื่อน pointer ของข้อมูลไปยัง record ที่อยู่ก่อนหน้า record ปัจจุบันใน Recordset

```
Datacontrol.RecordSet.MoveFirst
```

ตัวอย่างโปรแกรม

Dim newRecord As Boolean

Private Sub Form_Activate()

With Data1.Recordset

 .MoveLast

 .MoveFirst

End With

End Sub

Private Sub cmdAdd_Click()

newRecord = True

Data1.Recordset.AddNew

cmdAdd.Enabled = False

cmdDel.Enabled = True

End Sub

Private Sub CmdUpdate_Click()

With Data1.Recordset

If Not newRecord Then

 .Edit

End If

 .Update

newRecord = False

.Bookmark = .LastModified

End With

cmdAdd.Enabled = True

cmdDel.Enabled = True

End Sub

Private Sub cmdDel_Click()

If MsgBox("โปรดยืนยันการลบ", vbYesNo) = vbYes Then

 With Data1.Recordset

 .Delete

 .MoveNext

 If .EOF = True Then .MoveLast

 End With

End If

End Sub

Private Sub Data1_Reposition()

Dim CurrentRecord As Long, NumberOfRecord As Long

Dim dataCaption As String, status As Boolean

```

If newRecord = True Then
    dataCaption = "New Record"
    status = False
Else
    CurrentRecord = Str(Data1.Recordset.AbsolutePosition + 1)
    NumberOfRecord = Str(Data1.Recordset.RecordCount)
    dataCaption = "Record : " & CurrentRecord & " / " & NumberOfRecord
    status = True
End If
With Data1
    .Caption = dataCaption
    .Enabled = status
End With
End Sub

```

เพิ่ม Control Cancel ลงใน Form Demo 22

การใช้ method CancelUpdate ใช้ในกรณีที่ขกเลิกการเพิ่มและการแก้ไขข้อมูลใน Bound Control ก่อนที่จะบันทึกข้อมูลลงใน recordset ของ data control เป็นการขกเลิก method "Add New" หรือ "Edit" ก่อนที่จะมีการใช้ Method "Update"

```
datacontrol.Recordset.CancelUpdate
```

```

Private Sub cmdCancel_Click()
    Dim CurrentRecord As Long
    Dim NumberOfRecord As Long
    If MsgBox("โปรดยืนยันการขกเลิกการเพิ่ม/แก้ไขข้อมูล", vbYesNo, "message") = vbYes Then
        With Data1.Recordset
            If Not newRecord Then .Edit
            .CancelUpdate
        End With
    End If
    cmdAdd.Enabled = True
    cmdDel.Enabled = True
    Data1.Enabled = True
    newRecord = False
    CurrentRecord = Str(Data1.Recordset.AbsolutePosition + 1)
    NumberOfRecord = Str(Data1.Recordset.RecordCount)
    Data1.Caption = "Record : " & CurrentRecord & " / " & NumberOfRecord

```

```
End Sub
```

การสร้างฐานข้อมูลโดยใช้โปรแกรม

ชนิดของอ็อบเจกต์ในโมเดล DAO (Data Access Object)

DAO ประกอบด้วยอ็อบเจกต์ที่เป็นตัวแทนส่วนต่างๆ ของฐานข้อมูล เช่น เขตข้อมูล (Field) ที่อยู่ในตาราง จะถูกมองเป็นอ็อบเจกต์ตัวหนึ่ง หรือแม้แต่ตัวตารางเอง ก็จะมีอ็อบเจกต์ที่เป็นตัวแทนของตารางด้วยเช่นกัน ข้อผิดพลาด (error) ที่เกิดขึ้น ก็คืออ็อบเจกต์ตัวหนึ่งเช่นกัน ดังนั้นองค์ประกอบต่างๆ ที่อยู่ใ้ฐานข้อมูล จะถูก VB มองเป็นอ็อบเจกต์ รายละเอียดเบื้องต้น ของอ็อบเจกต์ DAO มีดังนี้

- **อ็อบเจกต์ DBEngine (DBEngine Object)** เป็นอ็อบเจกต์ระดับบนสุด เป็นตัวแทนของ JET Engine ที่ใช้สำหรับควบคุม และ/หรือจัดการ โครงสร้างของฐานข้อมูลที่สร้างขึ้นมา ซึ่งจะมีได้เพียง 1 ที่เท่านั้น ใน 1 โปรเจกต์
- **อ็อบเจกต์ Workspace (Workspace Object)** เป็นอ็อบเจกต์ที่ใช้สำหรับกำหนดพื้นที่สมมติขึ้นมา เพื่อเก็บฐานข้อมูลที่สร้างขึ้นมา
- **อ็อบเจกต์ Database (Database Object)** เป็นอ็อบเจกต์ที่เป็นตัวแทนของฐานข้อมูลนั้นๆ ซึ่งจะประกอบไปด้วย อ็อบเจกต์ที่เป็นองค์ประกอบของฐานข้อมูล ซึ่งผู้เขียนจะแบ่งออกเป็น 2 กลุ่มใหญ่ คือ

1. กลุ่มของอ็อบเจกต์ที่ใช้สำหรับกำหนดลักษณะฐานข้อมูลมี 4 ตัว คือ

- คอลเล็กชัน Workspaces (Workspaces Collection) ใช้สำหรับอ้างอิง หรือแทนกลุ่มของอ็อบเจกต์ Workspace
- คอลเล็กชัน TableDefs (TableDefs Collection) ใช้สำหรับอ้างอิง หรือแทนกลุ่มของอ็อบเจกต์ Table
- คอลเล็กชัน Fields (Fields Collection) ใช้สำหรับอ้างอิง หรือแทนกลุ่มของอ็อบเจกต์ Field
- คอลเล็กชัน Indexes (Indexes Collection) ใช้สำหรับอ้างอิง หรือแทนกลุ่มของอ็อบเจกต์ Index

2. อ็อบเจกต์ที่ใช้สำหรับกำหนดรายละเอียดอื่นๆ ของฐานข้อมูล มี 2 ตัว

- คอลเล็กชัน Relations (Relations Collection) ใช้สำหรับอ้างอิง หรือแทนกลุ่มของอ็อบเจกต์ Relation

- กอเล็กชัน QueryDefs (QueryDefs Collection) ใช้สำหรับอ้างอิง หรือแทนกลุ่มของอ็อบเจ็กต์ QueryDef

การใช้งานอ็อบเจ็กต์พื้นฐานข้อมูล

วิธีการนำอ็อบเจ็กต์ DAO มาใช้ในโปรเจ็กต์ ซึ่งมีขั้นตอนดังนี้

1. เลือกคำสั่ง Project/References จะได้ไอคอนบ็อกซ์ References
2. ให้เลือกตัวเลือก Microsoft DAO 3.6 Object Library

Demo23

Option Explicit

Private Sub cmdCreate_Click()

'Create data access object variables

Dim dbWorkspace As Workspace

Dim dbDatabase As Database

Dim dbTableDef As TableDef

Dim dbName As Field

Dim dbNumber As Field

'Create data access objects

Set dbWorkspace = DBEngine.Workspaces(0)

Set dbDatabase = dbWorkspace.CreateDatabase(_
"a:\PHONES.MDB", dbLangGeneral)

Set dbTableDef = dbDatabase.CreateTableDef("Phones")

Set dbName = dbTableDef.CreateField("Name", dbText)

Set dbNumber = dbTableDef.CreateField("Number", dbText)

'Set field properties

dbName.Size = 30

dbNumber.Size = 30

'Append each field object to its table object

dbTableDef.Fields.Append dbName

dbTableDef.Fields.Append dbNumber

'Append each table to its database


dbDatabase.TableDefs.Append dbTableDef

'Close completed database

dbDatabase.Close

MsgBox "Database Created", , "Create Phones"

End Sub



การเขียนโปรแกรมให้ติดต่อกับ database โดยไม่ใช้ Data Control

demo24

Option Explicit

'Create new data access object variables

Private dbWorkspace As Workspace

Private dbDatabase As Database

Private dbTable As Recordset

Private dbName As Field

Private dbNumber As Field

Private Sub cmdNext_Click()

'Move only if current record isn't a blank one

If txtName.Text <> " " And txtNumber.Text <> " " Then

'Save any changes to current record

UpdateRecord

'Move to next record

dbTable.MoveNext

'Prepare new record if at end of table

If dbTable.EOF Then NewRecord

'Display data from record

DisplayFields

End If

'Keep focus on text boxes

txtName.SetFocus

End Sub

Private Sub cmdPrevious_Click()

'Save any changes to current record

UpdateRecord

'Step back one record

dbTable.MovePrevious

'Don't go past first record

If dbTable.BOF Then dbTable.MoveNext


'Display data from record

DisplayFields

'Keep focus on text boxes

txtName.SetFocus

End Sub



Private Sub Form_Load()

```
'Create data access objects  
Set dbWorkspace = DBEngine.Workspaces(0)  
'Change database path if necessary  
Set dbDatabase = dbWorkspace.OpenDatabase("a:\PHONES.MDB")  
Set dbTable = dbDatabase.OpenRecordset("Phones", dbOpenTable)  
'Use special handling if new database  
If dbTable.BOF And dbTable.EOF Then NewRecord  
'Start on first record  
dbTable.MoveFirst  
'Display data from record  
DisplayFields
```

End Sub

Private Sub NewRecord()

```
'Add new record  
dbTable.AddNew  
'Install a space in each field  
dbTable!Name = " "  
dbTable!Number = " "  
'Update database  
dbTable.Update  
'Move to new record  
dbTable.MoveLast
```

End Sub

Private Sub UpdateRecord()

```
'Prepare table for editing  
dbTable.Edit  
'Copy text box contents into record fields  
dbTable!Name = txtName.Text  
dbTable!Number = txtNumber.Text  
'Update database  
dbTable.Update
```

End Sub

Private Sub DisplayFields()

```
'Display fields in text boxes  
txtName.Text = dbTable!Name  
txtNumber.Text = dbTable!Number
```

End Sub